Incidence, mortality and drug initiation in Danish T2 diabetes patients

SDC
March 2016
http://bendixcarstensen.com/DMreg/daffodil
Version 1

Compiled Wednesday 2nd March, 2016, 17:48
from: /home/bendix/sdc/proj/daffodil/R/incmort.tex

Bendix Carstensen Steno Diabetes Center, Gentofte, Denmark
& Department of Biostatistics, University of Copenhagen
bxc@steno.dk
http://BendixCarstensen.com

Contents

Chapter 1

Overview and data structure

1.1 Aims

The aim of this study is to assess the occurrence of
Specifically we shall use register data to assess

- incidence of T2 diabetes in Denmark:
  - non-pharmacologically treated
  - pharmacologically treated
- incidence of pharmacological treatment
- mortality of diabetes patients
- (for comparison) mortality of non-diabetes persons

Once these quantities have been described as functions of sex, (attained) age, calendar time, date of birth and — where relevant — duration of diabetes and treatment, we will compute derived measures:

- prevalence of T2 diabetes
- life-time lost to diabetes
- life-time spent with diabetes without pharmacological treatment
- life-time spent with diabetes with pharmacological treatment

These quantities will be quantified as functions of age and calendar time of diagnosis of diabetes. Also, the time-trends in these quantities will be described.

1.2 Concepts

In order to formalize and illustrate the quantities we are concerned with, we show the relevant transitions:

```
> library(Epi)
> TM <- matrix(NA,4,4)
> rownames(TM) <- colnames(TM) <- c("Well","DM-none","DM-drug","Dead")
> TM["Well","DM-none"] <-
+ TM["Well","DM-drug"] <-
+ TM["DM-none","DM-drug"] <-
+ TM["Well","Dead"] <-
+ TM["DM-none","Dead"] <-
+ TM["DM-drug","Dead"] <- 1
> wh <- list( x=c(10,85,85,60), y=c(50,90,10,50) )
> boxes( TM, boxpos=wh, hmult=3, wmult=1.5, col.bg=gray(c(1,1,1,0.8)) )
```



Figure 1.1: *Incidence and mortality transitions in the description of rates in diabetes occurrence. "DM-none" refers to patients diagnosed with diabetes but not on any antidiabetic drug, and "DM-drug" to patients that have filled at least one prescription for an antidiabetic drug.*

It is not necessarily of interest to model the incidence rates of "DM-none" and "DM-drug", the *sum* of the rates (total incidence of T2 diabetes) and the rate-ratio (essentially the odds of starting on pharmacological treatment) are more likely of interest. However, modeling of the two rates by age-period-cohort models [1] immediately yields age-period-cohort models for the sum and the ratio too.

Mortality rates can in the same vein be modeled separately or jointly for the two diabetes states, and again the sum or the ratio of the rates can be derived from the

separate models. Likewise with the ratio relative to the mortality rates of the non-diabetic part of the population ("Well").

1.3 Data structure and construction

The structure of the analysis data sets is basically one table of person years and types of events for each of the three transient states. The tables have different classifying sets; the "Well" dataset is classified by sex, age and calendar time; the "DM-none" by sex, age, calendar time and duration of DM and the "DM-drug" dataset by sex, age, calendar time, duration of DM and duration of drug treatment.

Data is constructed by tabulating events of DM and death, and the person-years spent in each state. The person-years in the "Well" state is obtained by taking the person-years in the entire Danish population and subtracting the person-years among those in the two DM states.

In the tabulations, age and calendar time is grouped in 1-year intervals, and time since diagnosis of DM, respectively drug initiation in 6-month intervals, the latter giving up to 40 different classes; table 1.1 gives an overview of the classifiers and outcome variables in the dataset.

Table 1.1: *Classifiers and outcome counters in the analysis data sets*

		Dataset (transient state)			
		Well	DM-none	DM-drug	Levels
Events:	DM-none	•			
	DM-drug	•	•		
	Dead	•	•	•	
Classifiers:	sex	•	•	•	2
	age	•	•	•	100
	cal.time	•	•	•	20
	DM dur.		•	•	40
	drug dur.			•	40
No. cells(1000s)		4	160	≈ 3000	

For the analysis of rates out of "Well" we will have $2 \times 100 \times 20 = 4,000$ entries, for the analysis of rates out of "DM-none" we will have $2 \times 100 \times 20 \times 40 = 160,000$ entries and for the analysis of rates out of "DM-drug" we will have (at most)
$2 \times 100 \times 20 \times 40 \times 40 = 6,400,000$. The latter is most likely a gross exaggeration, since we will hardly have all 1600 possible combinations of DM duration and drug duration, notably the latter is never larger than the former, so we will certainly have less then 3,200,000 records in that dataset.

Analysis of the rates is straight-forward using the number of events as response in Poisson-models with the log-person-years as offset and the effects of the time-scales as smooth parametric curves (natural splines).

1.4 SAS code

Here we devise the SAS-code to be used for the outlined analyses. Note that we first rename the variables from the mother dataset to names more mnemonically pertinent to the task at hand, and then make the Lexis split of data.

```
libname her "../data" ;
%inc "../../sas/Lexis.sas" ;

data her.base ;
  set her.personbase ( rename = ( doBth = DOBirth
                                  doDth = DoDeath
                                  doEnd = DoEndStudykorr
                                  doMed = DoRPMS
                                  doDM  = first_DM_korr ) ) ;
  doIni = "01JAN1995"d ; * Convenience constant ;
run ;

**************************************************************************
* Set up data for enumerating PY and events of different types from
* the DMnone state ;
data DMnone ;
  set her.base ;
  if ( doDM > .z and ( doDM < doMed or doMed le .z ) ) ;
  entry = max( doIni, doDM  ) ;
   exit = min( doEnd, doMed, doDth ) ;
 * fail is 2 if the exit is to medication, and 1 if it is to death ;
   fail = ( doMed > .z and doMed = exit ) * 2 +
          ( doDth > .z and doDth = exit ) ;
  if fail gt 2 then put "Hwa fanden:" fail= doMed= doEnd= doDth= ;
run ;

%Lexis( data = DMnone
         out = byP,
      breaks = 0 to 125 by 1,
      origin = "01JAN1900",
       scale = 365.25,
        left = P ) ;

data byP ; set byP ; P = P+1900 ; run ;

%Lexis( data = byP,
         out = byAP,
      breaks = 0 to 125 by 1,
      origin = doBth,
       scale = 365.25,
        left = A ) ;

%Lexis( data = byAP
         out = byAPD,
      breaks = 0 to 125 by 0.5,
      origin = doDM,
       scale = 365.25,
        left = dur,
        risk = Y ) ;

data byAPD ;
  set byAPD ;
  DMd = ( fail = 2 ) ;
  Dth = ( fail = 1 ) ;
run ;
```

```
proc summary  data=byAPD  nway ;
  class sex A P dur ;
  var Y DMd Dth ;
  output out = her.DMnone ( keep = sex A P dur Y DMd Dth )
         sum = ;
run ;

**************************************************************************
* Set up data for enumerating PY and events of different types from
* the DMdrug state ;
data DMdrug ;
  set her.base ;
  if ( doMed > .z ) ;
  entry = doMed ;
   exit = min( doEnd, doDth ) ;
   fail = ( doDth > .z and doDth = exit ) ;
run ;

%Lexis( data = DMdrug,
          out = byP,
       breaks = 0 to 125 by 1,
       origin = "01JAN1900",
        scale = 365.25,
         left = P ) ;

data byP ; set byP ; P = P+1900 ; run ;

%Lexis( data = byP
          out = byAP,
       breaks = 0 to 125 by 1,
       origin = doBth,
        scale = 365.25,
         left = A ) ;

%Lexis( data = byAP
          out = byAPD,
       breaks = 0 to 125 by 0.5,
       origin = doDM,
        scale = 365.25,
         left = dur ;

%Lexis( data = byAPD
          out = byAPDd,
       breaks = 0 to 125 by 0.5,
       origin = doMed,
        scale = 365.25,
         left = ddur,
         risk = Y ) ;

proc summary  data = byAPDd ( rename = ( fail=Dth ) )  nway ;
  class sex A P dur ddur ;
  var Y Dth ;
  output out = her.DMdrug ( keep = sex A P dur ddur Y Dth )
         sum = ;
run ;

**************************************************************************
* Enumerate the total deaths and risk time in DMnone and DMdrug for
* subtraction from the total Danish population data ;
data FU ;
  set her.DMnone ( keep = sex A P Y Dth )
```

```
      her.DMdrug ( keep = sex A P Y Dth ) ;
run ;

proc summary  data = FU  nway ;
  class sex A P ;
  var Y Dth ;
  output out = YDth ( keep = sex A P Y Dth )
         sum = ;
run ;

************************************************************************
* Set up data for enumerating events of different types from
* the Well state ;
data well ;
  set her.base ;
  DM  = ( doDM  > doIni & doDM < doMed ) ;
  DMd = ( doMed > doIni & doDM = doMed ) ;
    A = floor( ( doDM - doBth ) / 365.25 ) ;
    P = floor( ( doDM         ) / 365.25 ) ;
run ;

proc summary  data = well  nway ;
  class sex A P ;
  var DM DMd ;
  output out = DMnd ( keep = sex A P DM DMd )
         sum = ;
run ;

************************************************************************
* Merge the follow-up events in Well and the person-years and deaths and
* PY in the two DM states for subsequent subtraction from total
* population Y and D ;
data her.Well ;
  merge YDth /* variables  Y and Dth */
        DMnd /* variables DM and DMd */ ;
  by sex A P ;
run ;

************************************************************************
* Convert the relevant datasets to Xport datasets ;
options validvarnames=V6 ; * Not really necessary, all varnames short ;
libname xDMn xport "..\data\DMnone.xpt" ;
libname xDMd xport "..\data\DMdrug.xpt" ;
libname xWll xport "..\data\Well.xpt"   ;
proc copy in=her out=xDMn memtype=data ; select DMnone ; run ;
proc copy in=her out=xDMd memtype=data ; select DMdrug ; run ;
proc copy in=her out=xWll memtype=data ; select Well   ; run ;
```

References

[1] B Carstensen. Age-Period-Cohort models for the Lexis diagram. *Statistics in Medicine*, 26(15):3018–3045, July 2007.